

2 トリガー条件式

概要

トリガーで使用される条件式は非常に柔軟です。これらを使って、監視の解析結果に関する複雑な論理テストを作成することができます。

シンプルかつ便利な条件式の例を以下に記載します。

```
{<server>:<key>.<function>( <parameter> )}<operator><constant>
```

1 関数

トリガー関数を使用して、収集値、現在時刻および他の要素を参照することができます。 [サポートされている関数](#)のリストをご覧ください。

2 関数パラメータ

ほとんどの数値関数は、秒数をパラメータとして受け入れます。

プレフィクス#を使って、パラメータに異なる意味がある旨を記述することができます。

関数コール	意味
sum(600)	600秒内の全値の合計
sum(#5)	直近5個の値の合計

プリフィクス#を使用している値に異なる意味「n個前の値を選択させる」を持たせるので、直近3からさかのぼって値3、7、2、6、5が与えられている場合、**last(#2)** は7が返され、**last(#5)**は5が返されません。

パラメータを無視する関数に対しても、当該パラメータを付与する必要があります。例: **last(0)**

avg, count, last, minおよび**max**関数は、秒単位の **time_shift**パラメータをサポートします。このパラメータによって、過去の一定期間のデータを参照することができます。例えば、**avg(1h,1d)**は、1日前の1時間の平均値を返します。

トリガーは履歴情報の評価しか行いません。履歴が使用できない場合（特に、タイムシフト関連）には、トレンド情報が使用できないため、少なくともトリガー関数が要求する期間中は、履歴を保持する必要があります。

「300」秒の代わりに「5m」（5分）、「86400」秒の代わりに「1d」（1日）など、トリガー条件式でサポートされた [単位記号](#) を使用することができます。「1K」は「1024」バイトを意味します。

3 演算子

次の演算子がトリガーに対してサポートされます（優先度の高いものから順番に実行）

優先度	演算子	定義
1	/	割り算
2	*	掛け算
3	-	引き算
4	+	足し算
5	<	より小さい値。演算子は以下のように定義されます: $A < B \Leftrightarrow (A <= B - 0.000001)$
6	>	より大きい値。演算子は以下のように定義されます: $A > B \Leftrightarrow (A >= B + 0.000001)$
7	#	異なる値。演算子は以下のように定義されます: $A \# B \Leftrightarrow (A <= B - 0.000001) \mid (A >= B + 0.000001)$
8	=	同じ値。演算子は以下のように定義されます: $A = B \Leftrightarrow (A > B - 0.000001) \& (A < B + 0.000001)$
9	&	論理AND
10		論理OR

4 トリガー例

例 1

www.zabbix.com上では、プロセッサ負荷が高すぎます

```
{www.zabbix.com:system.cpu.load[all,avg1].last(0)}>5
```

[www.zabbix.com:system.cpu.load[all,avg1]]によって、監視したパラメータのショートネームが与えられます。これは、サーバが「www.zabbix.com」であり、監視対象キーが「system.cpu.load[all,avg1]」であることを表しています。関数last()を用いることによって、最新値を参照します。最後に、「>5」はwww.zabbix.comからの最新プロセッサ負荷の計測値が5よりも大きい時は、常に障害状態にあることを意味します。

例 2

www.zabbix.comは高負荷です

```
{www.zabbix.com:system.cpu.load[all,avg1].last(0)}>5 | {www.zabbix.com:system.cpu.load[all,avg1].min(10m)}>2
```

現在のプロセッサ負荷が5よりも大きい場合、または直近10分間のプロセッサ負荷が2よりも大きかった場合に、障害になります。

例 3

/etc/passwdが変更されました

関数diffの使用:

```
{www.zabbix.com:vfs.file.cksum[/etc/passwd].diff(0)}>0
```

/etc/passwdのチェックサム前の値が最新値と異なる場合に、障害になります。

/etc/passwd, /etc/inetd.conf, /kernelなどの重要ファイルの変更を監視する際には、同様の条件式が役に立ちます。

例 4

誰かがインターネットから大容量ファイルをダウンロードしています

関数minの使用:

```
{www.zabbix.com:net.if.in[eth0,bytes].min(5m)}>100K
```

直近5分以内に、eth0上の受信バイト数が100 KB以上の時に、障害になります。

例 5

クラスター化されたSMTPサーバの両ノードが停止しました

1つの条件式で2つの異なるホストを使用する際にはご注意ください:

```
{smtp1.zabbix.com:net.tcp.service[smtp].last(0)}=0&{smtp2.zabbix.com:net.tcp.service[smtp].last(0)}=0
```

smtp1.zabbix.comとsmtp2.zabbix.comの両方でSMTPサーバの両方が停止している時に、障害になります。

例 6

Zabbixエージェントをアップグレードする必要があります

関数strの使用:

```
{zabbix.zabbix.com:agent.version.str("beta8")}=1
```

Zabbixエージェントのバージョンがbeta8□恐らく1.0beta8□の場合に、障害になります。

例 7

サーバに接続できません

```
{zabbix.zabbix.com:icmpping.count(30m,0)}>5
```

直近30分間に、□zabbix.zabbix.com□が5回以上接続できなかった場合に、障害になります。

例 8

直近3分以内にハートビートがありません

関数nodataの使用:

```
{zabbix.zabbix.com:tick.nodata(3m)}=1
```

`tick`がタイプ `Zabbix trapper`を持つ必要があります。当該トリガーを機能させるために、アイテム `tick`を定義する必要があります。ホストからは`zabbix_sender`を用いて、当該パラメータに対してデータを定期的送信する必要があります。180秒以内にデータを受信しない場合には、トリガー値は「障害」となります。

例 9

夜間にCPUアクティビティが行われます

関数timeの使用:

```
{zabbix:system.cpu.load[all,avg1].min(5m)}>2&{zabbix:system.cpu.load[all,avg1].time(0)}>000000&{zabbix:system.cpu.load[all,avg1].time(0)}<060000
```

夜間(00:00-06:00)に限定して、トリガーによってそのステータスを障害に変えることができます。

例 10

クライアントのローカル時間がZabbixのサーバの時間と同期化されているかをチェックします

関数fuzzytimeの使用:

```
{MySQL_DB:system.localtime.fuzzytime(10)}=0
```

サーバのMySQL_DBとZabbixサーバ上のローカル時間が10秒以上ずれている場合に、トリガーは障害状態に変わります。

例 11

(2番目の`time_shift`パラメータを使用)本日の平均負荷と昨日の同じ時間の平均値を比較します

```
{server:system.cpu.load.avg(1h)}/{server:system.cpu.load.avg(1h,1d)}>2
```

直近1時間の平均負荷が、昨日の同じ時間の平均値を2倍以上になっている場合に、当該条件式が発動されます。

5 ヒステリシス

異なる状態に対して、異なる条件をトリガーにする必要がある場合があります。例えば、サーバ室の温度が20°Cを超えた場合に「障害」のトリガーが生じるようにして、15°C以下になるまでは「障害」の状態を維持します。

これを行うには、次のトリガーを定義します。

例 1

サーバ室の温度が高すぎます

```
{TRIGGER.VALUE}=0&{server:temp.last(0)}>20 |  
{TRIGGER.VALUE}=1&{server:temp.last(0)}>15)
```

マクロ {TRIGGER.VALUE} の使用方法にご注意ください。マクロは最新のトリガー値を返します。

例 2

空きディスク容量が少なすぎます

障害: 直近5分間で10GB以下

リカバリ: 直近10分間で40GB以上

```
{TRIGGER.VALUE}=0&{server:vfs.fs.size[/,free].max(5m)}<10G |  
{TRIGGER.VALUE}=1&{server:vfs.fs.size[/,free].min(10m)}<40G)
```

マクロ {TRIGGER.VALUE} の使用方法にご注意ください。マクロは最新のトリガー値を返します。

本ページは2014/08/05時点の原文を基にしておりますので、内容は必ずしも最新のものとは限りません。最新の情報は、英語版のZabbix2.2 マニュアルを参照してください。

From:

<https://www.zabbix.com/documentation/2.2/> - **Zabbix Documentation 2.2**

Permanent link:

<https://www.zabbix.com/documentation/2.2/jp/manual/config/triggers/expression>

Last update: **2014/10/10 02:39**

